# $Id: PKGBUILD 88638 2013-04-20 19:08:21Z arodseth $
# Maintainer: Alexander Rødseth <rodseth@gmail.com>
# Contributor: Roman Kyrylych <roman@archlinux.org>
# Contributor: Tom Killian <tomk@runbox.com>
# Contributor: Daniel J Griffiths <ghost1227@archlinux.us>

pkgname=libnewt
pkgver=0.52.15
pkgrel=1
_tclver=8.6
pkgdesc="Not Erik's Windowing Toolkit - text mode windowing with slang"
arch=('x86_64' 'i686')
url='https://fedorahosted.org/newt/'
license=('GPL')
depends=('slang' 'popt' 'gpm')
makedepends=("tcl>=$_tclver" 'python2')
optdepends=('tcl: whiptcl support' \
            'python2: libnewt api through _snack module')
options=('!makeflags')
source=("https://fedorahosted.org/releases/n/e/newt/newt-$pkgver.tar.gz")
sha256sums=('7a6151923e7a8a950f9a8a21668a5780d09b0f35f9d76a7ec606c71c35a0e241')

prepare() {
  cd "$srcdir/newt-$pkgver"
  
  sed -i "s:tcl8.4:tcl$_tclver:" Makefile.in
  sed -i 's:bin/python:bin/python2:' *.py
  echo '#define USE_INTERP_RESULT 1' >> config.h
}

build() {
  cd "$srcdir/newt-$pkgver"

  ./configure --prefix=/usr --with-gpm-support
  make
}

package() {
  cd "$srcdir/newt-$pkgver"
  
  make prefix="$pkgdir/usr" install
}

# vim:set ts=2 sw=2 et:
